

## 3. ANALIZA ZMIENNOPRĄDOWA

Programy służące do symulacji układów elektronicznych najczęściej używane są w celu określenia charakterystyki częstotliwościowej układu. Analiza taka na obliczeniu wszystkich prądów i napięć w układzie, który jest pobudzany wymuszeniami o przebiegu sinusoidalnym. Analiza przeprowadzana jest przy założeniu, że:

- Częstotliwości wszystkich wymuszeń są jednakowe.
- Sygnały w układzie są na tyle małe, że można pominąć wszystkie efekty nieliniowe. Nieliniowe charakterystyki elementów zastępowane są charakterystykami liniowymi aproksymującymi zachowanie elementu w pobliżu statycznego punktu pracy.
- Układ znajduje się w stanie ustalonym.

Stosując rachunek symboliczny [17] oblicza się nie tylko wartości amplitudy prądów i napięć, ale także ich wzajemne przesunięcia fazowe. Z tego typu analizą blisko związana jest analiza szumów. Przez szum rozumiemy chaotycznie generowany sygnał elektryczny. Szumy towarzyszą działaniu każdego układu elektronicznego. Przyczyny powstawania szumów to:

- Dyskretny charakter przepływu prądu elektrycznego. Ładunek nie przepływa w sposób ciągły lecz przenoszony jest przez elektrony.
- Chaotyczne ruchy termiczne elektronów.

Szumy powstające w układzie są zawsze na tyle małe, że można przyjąć założenie o liniowości układu. Źródła szumu opisuje się przez podanie widma mocy sygnału szumu w dziedzinie częstotliwości. Używając charakterystyk częstotliwościowych układu oblicza się widmo mocy szumów na wyjściu układu oraz widmo mocy szumów sprowadzonych na wejście układu [14].

### 3.1. Analiza w dziedzinie częstotliwości

Instrukcja zlecająca programowi PSpice przeprowadzenie analizy zmiennoprądowej może przyjąć jedną z trzech następujących postaci:

```
.AC LIN l_punktów _start _stop  
.AC OCT l_punktów _start _stop  
.AC DEC l_punktów _start _stop
```

Przykłady:

```
.AC LIN 100 1 100
```

.AC OCT 10 1K 16K

.AC DEC 20 1MEGHZ 100MEGHZ

Częstotliwość wszystkich źródeł w układzie zmieniana będzie od wartości **\_start** do wartości **\_stop**. Zmiany zachodzą liniowo lub logarytmicznie w zależności od tego, które ze słów kluczowych **LIN**, **OCT** lub **DEC** zostanie użyte.

**LIN** Częstotliwość źródeł zmieniana jest liniowo. W tym wypadku parametr **L\_punktów** oznacza całkowitą liczbę punktów, które znajdują się na wykresie.

**OCT** Częstotliwość źródeł zmieniana jest logarytmicznie co oktawę. Parametr **L\_punktów** oznacza liczbę punktów wykresu przypadającą na każdą oktawę.

**DEC** Częstotliwość źródeł zmieniana jest logarytmicznie co dekadę. Parametr **L\_punktów** oznacza liczbę punktów wykresu przypadającą na każdą dekadę.

### 3.1.1. Wymuszenia

Należy teraz przypomnieć to co powiedziano na stronie 9 o deklarowaniu wymuszeń dla analizy zmiennoprądowej. W przypadku siły elektromotorycznej deklaracja taka powinna wyglądać następująco:

VXXXXXXXX n+ n- AC \_moduł \_faza

W przypadku siły prądomotorycznej jest ona podobna:

IXXXXXXXXX n+ n- AC \_moduł \_faza

Przykłady:

I1 1 0 AC 1V

VCER 10 32 DC 5V AC 5V 30 PULSE(0 5)

Słowo kluczowe **AC** może współlistnieć w deklaracji źródła ze słowami kluczowymi DC, TRAN oraz specyfikacją przebiegu czasowego źródła. Po słowie kluczowym **AC** następuje wartość modułu wymuszenia (w voltach lub amperach), a następnie faza (w stopniach). Brak słowa kluczowego **AC** powoduje, że program PSpice przyjmuje wartość źródła równą zeru (AC 0 0). Jeżeli w deklaracji źródła pominięta zostanie wartość fazy program przyjmie, że jest równa zeru (np. AC 1V = AC 1V 0).

### 3.1.2. Modele elementów nieliniowych

Przed przystąpieniem do wykonania zmiennoprądowej analizy małosygnalowej obliczany jest statyczny punkt pracy układu. Dokonuje się tego w taki sam sposób, jak w przypadku instrukcji .OP. Następnie tworzy się liniowy, małosygnalowy model układu. Polega to na wykonaniu następujących operacji:


- Na podstawie obliczonego statycznego punktu pracy układu obliczane są wartości prądów i napięć dla każdego elementu nieliniowego.
- Charakterystyka każdego z elementów nieliniowych jest linearyzowana.

□ Usuwane są z układu wszystkie wymuszenia stałoprądowe.

W ten sposób uzyskuje się model liniowy, pozbawiony wymuszeń stałoprądowych. W sposób poprawny oddaje on zachowanie układu pod wpływem wymuszeń zmiennoprądowych o niewielkiej amplitudzie.

Przykład:

Charakterystyka statyczna diody półprzewodnikowej opisywana jest następującym równaniem:



Rys.21. Linearyzacja charakterystyki diody.

$$I(U) = I_s \cdot \left[ \exp\left(\frac{U}{N \cdot U_t}\right) - 1 \right] \quad (35)$$

gdzie:

- I prąd płynący przez diodę;
- U napięcie panujące na diodzie;
- $I_s$  prąd nasycenia diody;
- N bezwymiarowy współczynnik emisji;
- $U_t$  potencjał termiczny<sup>1</sup>.

Jeżeli napięcie panujące na diodzie zmieni swoją wartość od  $U_0$  do  $U$  to powoduje to zmianę wartości prądu od  $I_0$  do  $I$ . Rozwijając prawą stronę równania (35) w szereg Taylora dla  $U=U_0$  i pomijając wyrazy rzędu wyższego niż jeden (linearyzacja) otrzymujemy (Rys. 21):

$$I(U) = I_s \cdot \left[ \exp\left(\frac{U_0}{N \cdot U_t}\right) - 1 \right] + \frac{I_s}{N \cdot U_t} \cdot \exp\left(\frac{U_0}{N \cdot U_t}\right) \cdot (U - U_0) \quad (36)$$

Jeżeli składowe zmienne wymuszeń w układzie mają małą wartość to zmiany napięcia na diodzie  $\Delta U = U - U_0$  i prądu płynącego przez diodę  $\Delta I = I - I_0$  są małe. Zatem zamiast opisywać diodę charakterystyką nieliniową (35) można posłużyć się przybliżeniem liniowym (36). Równanie (36) opisuje równoległe połączenie zastępczej siły prądomotorycznej  $I_z$  o wartości:

<sup>1</sup>W temperaturze 27°C potencjał termiczny  $U_t \approx 26\text{mV}$ .


$$I_z = I_s \cdot \left[ \exp\left(\frac{U_0}{N \cdot U_t}\right) - 1 \right] - \frac{I_s}{N \cdot U_t} \cdot \exp\left(\frac{U_0}{N \cdot U_t}\right) \cdot U_0 \quad (37)$$

oraz zastępczej przewodności  $G_z$ :

$$G_z = \frac{I_s}{N \cdot U_t} \cdot \exp\left(\frac{U_0}{N \cdot U_t}\right) \quad (38)$$

Analiza zmiennoprądowa dotyczy tylko składowych zmiennych wymuszeń. Siła prądomotoryczna  $I_z$  jest zatem w tym modelu zbędna. W rezultacie liniowy, małosygnalowy model diody stanowi przewodność  $G_z$  dana wzorem (38) — Rys. 22.

Sposób postępowania w przypadku tworzenia liniowego, małosygnalowego modelu dla każdego innego elementu nieliniowego jest podobny.



**Rys.22.** Tworzenie modelu liniowego, małosygnalowego diody.

### 3.1.3. Wzmacniacz oporowy

Obliczyć wzmocnienie napięciowe, oraz impedancję wejściową układu wzmacniacza oporowego przedstawionego na Rys. 23. Obliczenia przeprowadzić dla częstotliwości zmieniającej się w przedziale od 100Hz do 100MHz.

Dane potrzebne do obliczeń przedstawione są poniżej.

```
ANALIZA WZMACNIACZA
.MODEL BC148 NPN(IS=67.34F XTI=3 EG=1.11 VAF=100 BF=116.1 NE=3.779
+ ISE=32.7N IKF=33.53M XTB=1.5 BR=1.926M NC=2 ISC=0 IKR=0 RC=1.5
+ CJC=4.929P VJC=.75 MJC=.3333 FC=.5 CJE=3.316P VJE=.75
+ MJE=.3333 TR=7.77U TF=107.3P ITF=.4 VTF=10 XTF=2 RB=10)

VCC 1 0 9V ;źródło napięcia zasilania

R1 1 2 145K ;opornik polaryzujący bazę tranzystora
R2 2 0 76K ;opornik polaryzujący bazę tranzystora
R3 1 3 1.2K ;opornik kolektorowy
R4 4 0 300 ;opornik emiterowy
C1 2 5 47N


Q1 3 2 4 BC148 ;deklaracja tranzystora bipolarnego

VIN 5 0 DC 0 AC 1 ;deklaracja źródła sygnału

.AC DEC 10 100 100MEGHZ ;instrukcja analizy ac
.PRINT AC IR(VIN) II(VIN) ;wyniki w postaci tabeli
.PLOT AC VM(3) VP(3) ;wyniki w postaci wykresu
.PROBE V(5) V(3) I(VIN) ;wyniki przekazane do programu probe
.END ;koniec
```

Na początku opisu obwodu znajdujemy bardzo szczegółową deklarację modelu tranzystora

BC148 zastosowanego w układzie. W deklaracji występuje wiele parametrów, tak że nie mieści się ona w jednej linii. Wobec tego kontynuowana jest w następnych liniach. Linia kontynuacji musi zaczynać się od znaku „+”. Każda deklaracja (instrukcja) może zostać zapisana za pomocą dowolnej liczby linii kontynuacji. Szczegółowe omówienie parametrów modelu tranzystora



Rys.23. Układ wzmacniacza oporowego.


odłożymy do rozdziału 6. Tutaj wymienimy tylko najważniejsze spośród tych, które znalazły się w przykładzie.

- Is** prąd nasycenia złącza tranzystora;
- Vaf** napięcie Early-ego;
- Bf** wzmocnienie prądowe podczas pracy normalnej;
- Br** wzmocnienie prądowe podczas pracy inwersyjnej;
- Rc** rezystancja kolektora;
- Cjc** pojemność złącza kolektor–baza przy zerowej polaryzacji;
- Vjc** potencjał wbudowany złącza kolektor–baza;
- Mjc** wykładnik opisujący profil domieszkowania złącza kolektor–baza;
- Cje** pojemność złącza emiter–baza przy zerowej polaryzacji;
- Vje** potencjał wbudowany złącza emiter–baza;
- Mje** wykładnik opisujący profil domieszkowania złącza emiter–baza;
- Tf** czas przelotu podczas pracy normalnej;
- Tr** czas przelotu podczas pracy inwersyjnej.

Na wejściu układu zadeklarowane zostało źródło napięcia, którego składowa stała (DC) jest równa zero, natomiast składowa zmienna (AC) ma amplitudę i fazę równą zero. Czy obliczenia dla tak dużej wartości sygnału zmiennego mają sens? Czy przy tak dużej wartości napięcia zmiennego nie są istotne w pracy tranzystora efekty nieliniowe? Przypomnijmy sobie definicję wzmocnienia napięciowego dla układu liniowego. Jeżeli źródło napięcia wejściowego jest w układzie jedynym źródłem napięcia to wzmocnieniem napięciowym  $K$  nazywamy stosunek napięcia na wyjściu układu  $U_{wy}$  do napięcia na wejściu  $U_{we}$ :

$$K = \frac{U_{wy}}{U_{we}} \quad (39)$$

Jeżeli więc przeprowadzimy analizę układu dla realnej wartości napięcia wejściowego to wzmacnienie obliczymy jako stosunek napięcia wyjściowego do wejściowego. Dla układu liniowego<sup>2</sup> wzmacnienie napięciowe nie jest zależne od wielkości napięcia wejściowego. A zatem im większe napięcie wejściowe tym większe napięcie wyjściowe. W szczególności jeżeli  $U_{we}=1[V]$  to napięcie wyjściowe, co do wartości, równe jest wzmacnieniu układu. Dzięki temu można uniknąć dzielenia koniecznego w przypadku, gdy  $U_{we} \neq 1[V]$ .




**Rys.24.** Moduł wzmacnienia układu wzmacniacza oporowego.

Na Rys. 24 przedstawiono obliczony przez program PSpice moduł wzmacnienia układu. Pamiętajmy, że kiedy stosujemy rachunek symboliczny napięcia i prądu w układzie są liczbami zespolonymi. Wobec tego wzmacnienie układu jest także liczbą zespoloną. Program graficzny PROBE symbol V(3) interpretuje jako moduł potencjału węzła 3<sup>3</sup>. Aby wykreślić fazę wzmacnienia, ilustrującą przesunięcie fazowe między napięciem wyjściowym, a wejściowym,

<sup>2</sup>Przed analizą AC każdy układ jest linearyzowany!

<sup>3</sup>Potencjał węzła 3 jest równy co do wartości wzmacnieniu układu.

należy dodać przyrostek P (ang. **p**hase — faza). W ten sposób uzyskujemy wyrażenie VP(3) — Rys. 25.



**Rys.25.** Przesunięcie fazowe wzmocnienia układu wzmacniacza oporowego.

Podobnie jak wzmocnienie rezystancja wejściowa układu jest opisywana liczbą zespoloną — mówi się o impedancji wejściowej układu. Część rzeczywista impedancji wejściowej to rezystancja wejściowa natomiast część urojona impedancji wejściowej to reaktancja wejściowa. Ponieważ napięcie wejściowe jest równe potencjałowi V(5) to impedancję wejściową można obliczyć jako stosunek potencjału V(5) do prądu płynącego przez źródło VIN ze znakiem minus<sup>4</sup>. Jeżeli uwzględnimy, że potencjał V(5)=1[V] dla wszystkich częstotliwości to rezystancję wejściową  $R_{we}$  będzie można obliczyć wg wzoru::


$$R_{we} = \frac{V(5) \cdot \text{Re}[-I(VIN)]}{I(VIN) \cdot \bar{I}(VIN)} \quad (40)$$

Podobnie można obliczyć wartość reaktancji wejściowej  $X_{we}$  :

<sup>4</sup>Dodatni prąd płynie od bieguna dodatniego źródła przez źródło do bieguna ujemnego.

$$X_{we} = \frac{V(5) \cdot \text{Im}[I(VIN)]}{I(VIN) \cdot \bar{I}(VIN)} \quad (41)$$

Na Rys. 26 przedstawiono zależność obu wymienionych wielkości od częstotliwości sygnału. Skorzystano przy tym z faktu, że część rzeczywistą prądu  $I(VIN)$  można otrzymać dodając przyrostek R (ang. r real part — część rzeczywista) natomiast część urojoną przez dodanie przyrostka I (ang. i imaginary part — część urojona).



**Rys.26.** Rezystancja i reaktancja wejściowa układu wzmacniacza oporowego.

### 3.1.4. Instrukcje wyprowadzania danych

Zbiór danych wejściowych dla programu PSpice, w którym opisano strukturę wzmacniacza oporowego (strona 48) zawiera dwie instrukcje .PRINT i .PLOT, o których nie było jeszcze mowy. Służą one do umieszczania wyników obliczeń w zbiorze wyjściowym. Pierwsza z nich powoduje umieszczenie w zbiorze wyjściowym tabeli, natomiast druga umieszcza tam wykres. Zbiór z danymi wyjściowymi jest tworzony tak aby mógł być wydrukowany za pomocą drukarki, która pracuje w trybie tekstowym.

Format instrukcji umieszczającej w zbiorze wyjściowym wyników w postaci tabeli jest następujący:

**.PRINT \_typ <\_lista>**



Przykłady:

```
.PRINT DC V(2) I(Vin) V(2,3)
.PRINT AC VM(2) VR(5,6)
.PRINT NOISE INOISE DB(ONoise)
.PRINT TRAN V(3) V(2,3)
```

Pole **\_typ** określa, dla której z analiz drukowane będą wyniki. W polu tym mogą znaleźć się następujące słowa kluczowe:

- DC** wyprowadzane będą dane dotyczące analizy charakterystyk statycznych;
- AC** wyprowadzane będą dane dotyczące małosygnałowej analizy zmiennoprądowej;
- NOISE** wyprowadzane będą dane dotyczące analizy szumów;
- TRAN** wyprowadzane będą dane dotyczące analizy stanów nieustalonych.

Po specyfikacji typu analizy następuje lista wielkości **<\_lista>**, które zostaną umieszczone w zbiorze wyjściowym. W przypadku analiz DC, AC i TRAN na liście tej może się znaleźć:

- V(n1[,n2])** różnica potencjałów między węzłem o numerze n1, a węzłem o numerze n2; jeżeli parametr n2 nie zostanie podany program przyjmie, że chodzi o węzeł masy;
- I(Vxxx)** prąd płynący przez źródło napięcia o nazwie Vxxx; prąd płynie przez źródło od bieguna dodatniego przez źródło do bieguna ujemnego.

W przypadku analizy AC za literą V lub I mogą znaleźć się przyrostki, które uściślają specyfikację wielkości, które program PSpice powinien wyprowadzić.

- R** część rzeczywista;
- I** część urojona;
- M** moduł wielkości zespolonej;
- P** faza wielkości zespolonej;
- DB** moduł wielkości zespolonej wyrażony w decybelach —  $20 \cdot \log_{10}()$ .

W przypadku analizy szumów NOISE na liście wyjściowej mogą znaleźć się następujące słowa kluczowe:

- ONoise** szum całkowity na wyjściu układu (wyjście określone jest w instrukcji analizy szumów);
- INoise** szum odniesiony do źródła na wejściu układu (źródło wejściowe określone jest w instrukcji analizy szumów);
- DB(ONoise)** szum całkowity na wyjściu układu w decybelach — poziom odniesienia 1 [V]/[Hz]<sup>1/2</sup>;
- DB(INoise)** szum, w decybelach, odniesiony do źródła wejściowego — poziom odniesienia 1 [A]/[Hz]<sup>1/2</sup>.

W przypadku analizy wzmacniacza oporowego instrukcja .PRINT umieszczona w zbiorze danych wejściowych zleca programowi PSpice wyprowadzenie części rzeczywistej i części

urojonej prądu płynącego przez źródło wejściowe VIN. Tabela utworzona przez program PSpice przedstawiona jest poniżej.

\*\*\*\*\* 08/27/91 \*\*\*\*\* Evaluation PSpice (Jan. 1988) \*\*\*\*\* 13:05:16 \*\*\*\*\*

ANALIZA WZMACNIACZA

\*\*\*\* AC ANALYSIS

TEMPERATURE = 27.000 DEG C

\*\*\*\*\*

FREQ	IR(Vin)	II(Vin)
1.000E+02	-1.191E-05	-2.349E-05
1.259E+02	-1.686E-05	-2.642E-05
1.585E+02	-2.285E-05	-2.844E-05
1.995E+02	-2.945E-05	-2.913E-05
2.512E+02	-3.602E-05	-2.830E-05
3.162E+02	-4.192E-05	-2.617E-05
3.981E+02	-4.675E-05	-2.319E-05
5.012E+02	-5.042E-05	-1.988E-05
6.310E+02	-5.305E-05	-1.663E-05
7.943E+02	-5.485E-05	-1.368E-05
1.000E+03	-5.605E-05	-1.113E-05
1.259E+03	-5.684E-05	-9.002E-06
1.585E+03	-5.734E-05	-7.259E-06
1.995E+03	-5.767E-05	-5.855E-06
2.512E+03	-5.787E-05	-4.738E-06
3.162E+03	-5.800E-05	-3.862E-06
3.981E+03	-5.809E-05	-3.185E-06
5.012E+03	-5.814E-05	-2.675E-06
6.310E+03	-5.817E-05	-2.306E-06
7.943E+03	-5.819E-05	-2.059E-06
1.000E+04	-5.821E-05	-1.921E-06
1.259E+04	-5.822E-05	-1.885E-06
1.585E+04	-5.822E-05	-1.949E-06
1.995E+04	-5.823E-05	-2.117E-06
2.512E+04	-5.823E-05	-2.398E-06
3.162E+04	-5.823E-05	-2.807E-06
3.981E+04	-5.823E-05	-3.365E-06
5.012E+04	-5.823E-05	-4.102E-06
6.310E+04	-5.824E-05	-5.057E-06
7.943E+04	-5.824E-05	-6.282E-06
1.000E+05	-5.824E-05	-7.842E-06
1.259E+05	-5.825E-05	-9.819E-06
1.585E+05	-5.827E-05	-1.232E-05
1.995E+05	-5.829E-05	-1.547E-05
2.512E+05	-5.832E-05	-1.945E-05
3.162E+05	-5.838E-05	-2.447E-05
3.981E+05	-5.847E-05	-3.079E-05
5.012E+05	-5.861E-05	-3.875E-05
6.310E+05	-5.882E-05	-4.877E-05
7.943E+05	-5.917E-05	-6.138E-05
1.000E+06	-5.972E-05	-7.725E-05
1.259E+06	-6.059E-05	-9.723E-05
1.585E+06	-6.198E-05	-1.224E-04
1.995E+06	-6.416E-05	-1.540E-04
2.512E+06	-6.763E-05	-1.936E-04
3.162E+06	-7.310E-05	-2.434E-04
3.981E+06	-8.175E-05	-3.058E-04
5.012E+06	-9.538E-05	-3.837E-04
6.310E+06	-1.168E-04	-4.804E-04
7.943E+06	-1.502E-04	-5.997E-04
1.000E+07	-2.021E-04	-7.449E-04
1.259E+07	-2.816E-04	-9.185E-04
1.585E+07	-4.011E-04	-1.120E-03
1.995E+07	-5.757E-04	-1.342E-03
2.512E+07	-8.207E-04	-1.571E-03
3.162E+07	-1.146E-03	-1.780E-03
3.981E+07	-1.545E-03	-1.933E-03
5.012E+07	-1.995E-03	-1.999E-03
6.310E+07	-2.450E-03	-1.961E-03
7.943E+07	-2.868E-03	-1.829E-03
1.000E+08	-3.215E-03	-1.631E-03

Instrukcja .PLOT, służąca do tworzenia wykresów, została przewidziana w standardzie SPICE2 w czasach, gdy drukarki graficzne były niezwykle drogie. Stąd instrukcja ta tworzy wykresy za pomocą standardowych znaków, które mogą być drukowane za pomocą drukarki wierszowej. Format instrukcji jest następujący:

**.PLOT \_typ <\_lista>**

Przykłady:

```
.PLOT DC V(2) I(Vin) V(2,3)
.PLOT AC VM(2) VI(5,6)
.PLOT NOISE INOISE DB(ONNOISE)
.PLOT TRAN V(3) V(2,3) -5,5
```

Parametr **\_typ** określa rodzaj analizy, której dotyczy instrukcja. Parametr ten przyjmuje takie same wartości jak w przypadku instrukcji .PRINT (patrz strona 53). Po specyfikacji typu analizy następuje lista wielkości **<\_lista>**, które zostaną umieszczone na wykresie. Może się na niej znaleźć co najwyżej 8 wielkości. Za każdą z nich można podać najmniejszą i największą wartość jaka znajdzie się na osi odciętych. Zatem **<\_lista>** składa się z pól w postaci:

**\_nazwa [\_max, \_min]**

Nazwy wielkości, które mogą znaleźć się w polu **\_nazwa** podlegają tym samym regułom jakim podlegają nazwy wielkości, które mogą wystąpić w instrukcji .PRINT (patrz strona 53). Parametr **\_min** określa najmniejszą wartość, która znajdzie się na osi odciętych, natomiast parametr **\_max** oznacza największą wartość, która znajdzie się na tej osi. Wartości obu parametrów dotyczą wszystkich wielkości, które wymieniono na lewo od nich. Jeżeli nie zostaną podane wartości **\_min** i **\_max** program wyznaczy je automatycznie. Jeśli zajdzie taka potrzeba na osi rzędnych znajdzie się kilka skal.

W przypadku wzmacniacza oporowego, analizowanego w poprzednim podpunkcie, umieszczenie w zbiorze danych wejściowych instrukcji .PLOT daje wykres modułu i fazy wzmocnienia przedstawiony poniżej.

```
***** 08/27/91 ***** Evaluation PSpice (Jan. 1988) ***** 13:05:16 *****
ANALIZA WZMACNIACZA
****      AC ANALYSIS                      TEMPERATURE = 27.000 DEG C
*****

LEGEND:
*: VM(3)
+: VP(3)

      FREQ          VM(3)
(*)-----          1.0000E+00  1.0000E+01  1.0000E+02  1.0000E+03  1.0000E+04
(+)-----          -2.0000E+02 -1.0000E+02  0.0000E+00  1.0000E+02  2.0000E+02

      1.000E+02  1.709E+00  .  *  +  .  .  .  .
      1.259E+02  2.033E+00  .  *  +  .  .  .  .
```

```

1.585E+02 2.367E+00 . * + . . . .
1.995E+02 2.687E+00 . * + . . . .
2.512E+02 2.972E+00 . *+ . . . .
3.162E+02 3.206E+00 . X . . . .
3.981E+02 3.386E+00 . +* . . . .
5.012E+02 3.516E+00 . + * . . . .
6.310E+02 3.606E+00 . + * . . . .
7.943E+02 3.667E+00 . + * . . . .
1.000E+03 3.707E+00 . + * . . . .
1.259E+03 3.733E+00 . + * . . . .
1.585E+03 3.749E+00 . + * . . . .
1.995E+03 3.760E+00 . + * . . . .
2.512E+03 3.767E+00 . + * . . . .
3.162E+03 3.771E+00 . + * . . . .
3.981E+03 3.774E+00 . + * . . . .
5.012E+03 3.775E+00 . + * . . . .
6.310E+03 3.776E+00 . + * . . . .
7.943E+03 3.777E+00 . + * . . . .
1.000E+04 3.777E+00 . + * . . . .
1.259E+04 3.778E+00 . + * . . . .
1.585E+04 3.778E+00 . + * . . . .
1.995E+04 3.778E+00 . + * . . . .
2.512E+04 3.778E+00 . + * . . . .
3.162E+04 3.778E+00 . + * . . . .
3.981E+04 3.778E+00 . + * . . . .
5.012E+04 3.778E+00 . + * . . . .
6.310E+04 3.778E+00 . + * . . . .
7.943E+04 3.778E+00 . + * . . . .
1.000E+05 3.778E+00 . * . . . + .
1.259E+05 3.778E+00 . * . . . + .
1.585E+05 3.778E+00 . * . . . + .
1.995E+05 3.778E+00 . * . . . + .
2.512E+05 3.778E+00 . * . . . + .
3.162E+05 3.778E+00 . * . . . + .
3.981E+05 3.778E+00 . * . . . + .
5.012E+05 3.778E+00 . * . . . + .
6.310E+05 3.778E+00 . * . . . + .
7.943E+05 3.778E+00 . * . . . + .
1.000E+06 3.778E+00 . * . . . + .
1.259E+06 3.777E+00 . * . . . + .
1.585E+06 3.777E+00 . * . . . + .
1.995E+06 3.776E+00 . * . . . + .
2.512E+06 3.774E+00 . * . . . + .
3.162E+06 3.772E+00 . * . . . + .
3.981E+06 3.768E+00 . * . . . + .
5.012E+06 3.762E+00 . * . . . + .
6.310E+06 3.753E+00 . * . . . + .
7.943E+06 3.738E+00 . * . . . + .
1.000E+07 3.716E+00 . * . . . + .
1.259E+07 3.681E+00 . * . . . + .
1.585E+07 3.628E+00 . * . . . + .
1.995E+07 3.549E+00 . * . . . + .
2.512E+07 3.435E+00 . * . . . + .
3.162E+07 3.278E+00 . * . . . + .
3.981E+07 3.073E+00 . * . . . + .
5.012E+07 2.826E+00 . * . . . + .
6.310E+07 2.550E+00 . * . . . + .
7.943E+07 2.269E+00 . * . . . + .
1.000E+08 2.004E+00 . * . . . + .
-----

```

Instrukcje `.PRINT` i `.PLOT` są rzadko wykorzystywane w przypadku programu PSpice. Znacznie częściej używa się instrukcji `.PROBE` powodującej przekazanie wyników obliczeń do programu graficznego PROBE. Format tej instrukcji jest następujący:


**`.PROBE <_lista>`**

Przykłady:

`.PROBE I(VII) V(5,2)`

`.PROBE VDB(5,3)`

Wielkości, które zostaną przekazane do programu PROBE umieszcza się na liście `<lista>`. Może się na niej znaleźć każda wielkość dopuszczalna przez instrukcję `.PRINT` (strona 52). W przypadku analizy zmiennoprądowej dopuszczalne jest stosowanie przyrostków (R,I,M,P,DB). Nie mają one jednak żadnego znaczenia. Za każdym razem do programu PROBE przekazane zostaną wyniki obliczeń w postaci liczb zespolonych. Natomiast Program PROBE służy do prezentacji wyników obliczeń w postaci graficznej oraz pozwala na dalsze



**Rys.27.** Sposób rozcięcia pętli sprzężenia zwrotnego generatora. A) Obwód elektryczny. B) Schemat blokowy.

ich przetwarzanie. W szczególności może być obliczona część rzeczywista, część urojona, moduł i faza przebiegu (patrz dodatek C). Program PROBE wykorzystuje możliwości graficzne komputera oraz posiadanej drukarki.

### 3.1.5. Analiza generatora

Możliwości jakie daje program PSpice w zakresie małosygnałowej analizy zmiennoprądowej można rozszerzyć o technikę umożliwiającą skuteczną analizę układów generacyjnych. W modelu generatora należy odszukać źródło sterowane i w tym miejscu przeciąć pętlę

sprężenia zwrotnego<sup>5</sup>. W przypadku źródła prądu sterowanego napięciem zabieg ten ilustruje Rys. 27. Układ jest niestabilny (jest generatorem), jeżeli dla częstotliwości generacji  $\omega_0$  transmitancja układu w otwartej pętli sprężenia zwrotnego  $K(i \cdot \omega)$  spełnia następujący warunek:

$$\begin{cases} \sqrt{K(i \cdot \omega_0) \cdot \overline{K(i \cdot \omega_0)}} \geq 1 \\ \arg[K(i \cdot \omega_0)] = 0 \end{cases} \quad (42)$$


Warunek ten wydaje się intuicyjnie zrozumiały. Generacja następuje w przypadku, gdy sygnał po przejściu przez układ znajdzie się na jego wejściu wzmacniony i w tej samej fazie. Ścisłe uzasadnienie tego warunku można znaleźć w książkach poświęconych układom automatyki [34],[12]. Należy zatem dokonać analizy AC układu generatora, w którym rozcięto pętlę sprężenia zwrotnego i znaleźć częstotliwość, dla której spełniony jest warunek (42).

Przykład:

Obliczyć częstotliwość drgań oraz dobroć układu generatora Collpitsa–a przedstawionego na Rys. 28. Układ wzmacniacza oporowego jest identyczny jak ten, który przedstawiono w paragrafie 3.1.3.

Jedynym źródłem sterowanym w analizowanym układzie, jest źródło prądu sterowane napięciem ukryte wewnątrz małosygnalowego modelu tranzystora.

Aby rozciąć pętlę sprężenia zwrotnego należy najpierw odtworzyć ten model. W tym celu za pomocą programu PSpice obliczymy statyczny punkt pracy układu. Potrzebne w tym celu dane przedstawione są poniżej.



Rys.28. Układ generatora Collpitsa-a.

```
ANALIZA GENERATORA - statyczny punkt pracy
.MODEL BC148 NPN(Is=67.34f Xti=3 Eg=1.11 Vaf=100 Bf=116.1 Ne=3.779
+ Ise=32.7n Ikf=33.53m Xtb=1.5 Br=1.926m Nc=2 Isc=0 Ikr=0 Rc=1.5
+ Cjc=4.929p Vjc=.75 Mjc=.3333 Fc=.5 Cje=3.316p Vje=.75
+ Mje=.3333 Tr=7.77u Tf=107.3p Itf=.4 Vtf=10 Xtf=2 Rb=10)
```

```
VCC 1 0 9V ;źródło napięcia zasilania
R1 1 2 145k ;opornik polaryzujący bazę tranzystora
R2 2 0 76k ;opornik polaryzujący bazę tranzystora
R3 1 3 1.2k ;opornik kolektorowy
R4 4 0 300 ;opornik emiterowy
```

<sup>5</sup>Rozcięcie pętli sprężenia zwrotnego w miejscu, gdzie występuje źródło sterowane nie powoduje zmiany warunków, w których układ pracuje.


Q1 3 2 4 BC241 ;deklaracja tranzystora bipolarnego

.OP  
.END

Po zakończonej analizie w zbiorze wyjściowym znajdziemy dane dotyczące małosygnalowych parametrów tranzystora.

```
***** 08/27/91 ***** Evaluation PSpice (Jan. 1988) ***** 15:40:02 *****
ANALIZA GENERATORA - statyczny punkt pracy
**** OPERATING POINT INFORMATION TEMPERATURE = 27.000 DEG C
*****
**** BIPOLAR JUNCTION TRANSISTORS
```

```
NAME      Q1
MODEL     BC148
IB        3.70E-05
IC        2.04E-03
VBE       6.25E-01
VBC       -5.30E+00
VCE       5.93E+00
BETADC    5.51E+01
GM        7.48E-02
RPI       1.13E+03
RX        1.00E+01
RO        5.16E+04
CBE       1.31E-11
CBC       2.46E-12
CBX       0.00E+00
CJS       0.00E+00
BETAAC    8.48E+01
FT        7.63E+08
```



Rys.29. Małosygnalowy model tranzystora.

Parametry potrzebne do odtworzenia modelu małosygnalowego tranzystora to:

- GM** transkonduktancja źródła sterowanego;
- RPI** oporność wejściowa tranzystora widziana z zacisków baza–emiter ( $1/y_{11}$ );
- RX** oporność obszaru bazy;
- RO** oporność widziana między zaciskami kolektora i emitera ( $1/y_{22}$ );
- CBE** pojemność złączowa baza–emiter;
- CBC** pojemność złączowa baza–kolektor.

Schemat elektryczny małosygnalowego modelu tranzystora przedstawiony jest na Rys. 29. Źródło prądu sterowane jest napięciem V(40,4). Aby rozciąć pętlę sprzężenia zwrotnego w układzie generatora wystarczy wprowadzić dodatkowy węzeł o numerze 5 i spowodować, aby źródło sterowane było potencjałem V(5). Układ będzie generował drgania o takiej częstotliwości, dla której transmitancja od potencjału V(5) do napięcia V(40,4) będzie liczbą rzeczywistą nie mniejszą od jedności. Potrzebną nam transmitancję obliczymy za pomocą programu PSpice. Dane przedstawiono poniżej.

## ANALIZA GENERATORA - CZĘSTOTLIWOŚĆ DRGAŃ

```

*model małosygnalowy tranzystora
RX 2 40 10           ;oporność obszaru bazy
RPI 40 4 1.13K      ;oporność dynamiczna złącza baza-emiter
RO 3 4 51.6K        ;oporność dynamiczna widziana między kolektorem i emiterem
CBE 40 4 13.1P      ;pojemność dynamiczna baza-emiter
CBC 40 3 2.46P      ;pojemność dynamiczna baza-kolektor
GM 3 4 5 0 7.48E-2  ;transkonduktancja tranzystora

*reszta wzmacniacza
R1 0 2 145K         ;oporniki polaryzujące bazę
R2 2 0 76K          ;oporność kolektorowa
R3 0 3 1.2K         ;oporność emiterowa
R4 4 0 300          ;oporność emiterowa

*elementy ustalające częstotliwość drgań
L1 6 2 10M
C1 6 0 10N
C2 2 0 3N
C3 3 6 100N

*źródło sygnału
RIN 5 0 1MEG        ;źródło sygnału
VIN 5 0 AC 1        ;do każdego węzła muszą być dołączone conajmniej dwa elementy

.AC DEC 200 1K 100MEG ;analiza w dziedzinie częstotliwości
.PROBE V(40,4)        ;wyniki przekazać do PROBE
.END                  ;koniec

```

Do sztucznie utworzonego węzła 5 dołączona jest siła elektromotoryczna VIN o wartości 1[V]. Dodatkowo do węzła 5 dołączona jest oporność RIN. Nie zmienia ona wartości potencjałów w obwodzie natomiast dzięki niej węzeł 5 nie jest węzłem „wiszącym w powietrzu” (do każdego węzła muszą być dołączone co najmniej dwa elementy — strona 5). Wzmocnienie układu jest równe co do wartości napięciu V(40,4).

Sprawdźmy najpierw czy nasz układ może być rzeczywiście generatorem. W tym celu wykreślamy za pomocą programu PROBE wzmocnienie układu na płaszczyźnie zespolonej (Rys. 30). Jest to możliwe dzięki temu, że program PROBE pozwala na opisanie osi poziomej wykresu za pomocą dowolnego wyrażenia. W szczególności może to być część rzeczywista napięcia V(40,4)<sup>6</sup> (które jest równe wzmocnieniu układu). Jeżeli na osi pionowej będziemy odkładać wartości części urojonej napięcia V(40,4) uzyskamy charakterystykę układu na płaszczyźnie zespolonej. Na Rys. 30 widać, że w punkcie, w którym charakterystyka przecina półprostą:


$$\operatorname{Re}(s) > 0, \operatorname{Im}(s) = 0$$

wartość wzmocnienia układu z pewnością jest większa od jedności. Istnieje zatem częstotliwość, dla której spełniony jest warunek (42).

Aby znaleźć częstotliwość generacji wykreślmy zależność fazy V(40,4) od częstotliwości (Rys. 31). Posługując się kursorem, który udostępnia program PROBE, można odczytać, że dla częstotliwości  $f_1 = 36,31$ [KHz] faza wzmocnienia wynosi  $\phi_1 = 2.764^\circ$  natomiast dla

<sup>6</sup>Należy pamiętać, że oś pozioma musi być wyskalowana w skali liniowej.





**Rys.30.** Oś pozioma: część rzeczywista wzmocnienia układu generatora w otwartej pętli sprzężenia zwrotnego. Oś pionowa: część urojona wzmocnienia.

częstotliwości  $f_2=36,73[\text{kHz}]$  faza  $\varphi_2=-0.7481^\circ$ . Częstotliwość  $f_0$ , dla której faza przyjmuje wartość zero, obliczamy stosując metodę interpolacji liniowej:

$$f_0 = f_1 - \varphi_1 \cdot (f_2 - f_1) / (\varphi_2 - \varphi_1) \approx 36.64 [\text{kHz}] \quad (43)$$


Ze względu na dokładność obliczeń wynik ten rozsądnie jest zaokrąglić do wartości  $f_0=36,6[\text{kHz}]$ .

Kolejnym zadaniem, które należy wykonać jest obliczenie dobroci generatora. Jest ona definiowana za pomocą następującego wzoru [35], [25]:

$$Q = \left[ -\frac{\omega}{2} \cdot \frac{\partial \arg[K(i \cdot \omega)]}{\partial \omega} \right]_{\omega=\omega_0} \quad (44)$$

Wielkość ta opisuje stabilność częstotliwości generowanych drgań. Ponieważ warunek na moduł transmitancji  $K(\omega)$  ma postać nierówności, o stabilności decyduje wrażliwość fazy transmitancji  $\arg[K(i \cdot \omega)]$  na zmiany pulsacji  $\omega$ . Jeżeli wrażliwość<sup>7</sup> ta jest duża to duże

<sup>7</sup>Zauważmy, że dobroć generatora jest definiowana podobnie do wrażliwości.



**Rys.31.** Faza wzmocnienia generatora w otwartej pętli sprzężenia zwrotnego.


zmiany parametrów układu zostaną skompensowane niewielką zmianą częstotliwości drgań tak aby warunek fazy został zachowany.

Wielkość występująca po prawej stronie wzoru (44) może być łatwo wykreślona za pomocą programu PROBE — Rys. 32. Częstotliwość jest identyfikowana przez nazwę FREQUENCY (ang.frequency — częstotliwość). Dobroć generatora odczytana z wykresu (za pomocą kursora) wynosi około  $154[\text{deg}] \cdot [\text{Hz}/\text{Hz}]$ .

Przedstawiona technika może być użyteczna w przypadku analizy układu generacyjnego, którego model jest bardzo skomplikowany, co wyklucza metody „ręczne”. W pracy [25], dla układu generatora pracującego w zakresie mikrofal, autor porównał wyniki obliczeń za pomocą symulatora z wynikami uzyskanymi doświadczalnie. Różnica, nie przekraczała 2%, tak dla częstotliwości generacji jak i dla dobroci układu. Świadczy o dużej użyteczności praktycznej przedstawionej metody.

### 3.2. Analiza szumów

Moduł programu PSpice przeznaczony do małosygnałowej analizy zmiennoprądowej przystosowany jest także do analizy szumów. Szумы termiczne generowane są przez każdy opornik w układzie. Podobnie szумы termiczne i szумы migotania ( $1/f$ ) generowane są przez



**Rys.32.** Dobroć układu generatora.

każdy przyrząd półprzewodnikowy. Każde ze źródeł szumu opisywane jest gęstością widmową mocy sygnału szumu. Poszczególne źródła szumów są w układzie elektronicznym nieskorelowane, dzięki czemu można obliczać wpływ każdego ze źródeł na szum wyjściowy układu osobno, a następnie zsumować poszczególne przyczynki [14]. Program PSpice oblicza, w podanym przedziale częstotliwości, gęstość widmową sygnału szumu na wyjściu układu wyrażoną w  $[V]/[Hz]^{1/2}$  lub w  $[A]/[Hz]^{1/2}$ , w zależności od charakteru wyjścia (napięciowe lub prądowe). Obliczana jest także gęstość widmowa szumów zredukowana do wejścia układu — gęstość sygnału szumu na wyjściu podzielona przez moduł odpowiedniej transmitancji układu.

### 3.2.1. Modele szumowe elementów [24]

Wszystkie modele szumowe elementów dostępne w programie PSpice utworzone są z kilku podstawowych elementów. Jest to:

- Model szumowy opornika. Składa się na niego niezależne źródło prądu połączone równolegle z opornikiem  $R$ . Wartość gęstości widmowej mocy sygnału szumu ( $I_{sz,R}$ )<sup>2</sup> określa następujący wzór:

$$I_{sz,R}^2 = \frac{4 \cdot k \cdot T}{R} \quad (45)$$

gdzie:

- R wartość oporności;
- k stała Boltzmanna =  $1,38 \cdot 10^{-23}$  [J/K];
- T temperatura w skali bezwzględnej.

Wzór ten opisuje szумы wynikające z bezładnych ruchów termicznych wykonywanych przez elektrony wewnątrz opornika.

- Model szumowy idealnego złącza p–n. Składa się z połączonych równolegle niezależnego źródła prądu, liniowej przewodności  $G_z$  równej wartości przewodności różniczkowej złącza  $G_z = (\partial I_z / \partial U_z)$  i liniowej pojemności  $C_z$  równej pojemności różniczkowej złącza  $C_z = (\partial Q_z / \partial U_z)$ . Gęstość widmowa mocy sygnału szumu  $(I_{sz,z})^2$  określona jest wzorem:

$$I_{sz,z}^2 = 2 \cdot q \cdot I_z + \frac{KF \cdot I_z^{AF}}{f} \quad (46)$$

gdzie:

- $I_z$  statyczny prąd płynący przez złącze;
- f częstotliwość;
- KF współczynnik szumów migotania;
- AF wykładnik szumów migotania;
- q ładunek elementarny =  $1,61 \cdot 10^{-19}$  [A·s].

Pierwszy składnik we wzorze (46) opisuje szum śrutowy złącza, którego przyczyną jest ziarnistość ładunku. Drugi składnik opisuje tzw. szum migotania, którego istota nie została do dzisiaj rozstrzygnięta.

- Model szumowy kanału tranzystora polowego. Składa się on z połączonych równolegle niezależnej siły prądomotorycznej i liniowego źródła prądu sterowanego prądem o wartości równej małosygnalowej transkonduktancji tranzystora  $G_m = \partial I_k / \partial U_{gs}$ . Gęstość widmowa mocy sygnału szumu  $(I_{sz,k})^2$  określa wzór:

$$I_{sz,k}^2 = \frac{8 \cdot k \cdot T \cdot G_m}{3} + \frac{KF \cdot I_k^{AF}}{f} \quad (47)$$

gdzie:

- $I_k$  statyczny prąd płynący przez kanał tranzystora FET;
- KF współczynnik szumów migotania;
- AF wykładnik szumów migotania;
- f częstotliwość;
- k stała Boltzmanna =  $1,38 \cdot 10^{-23}$  [J/K];
- T temperatura bezwzględna.

Powyższy wzór opisuje szумы termiczne w kanale (składnik stały) i szумы migotania (składnik odwrotnie proporcjonalny do częstotliwości).

Program PSpice buduje model szumowy rzeczywistego przyrządu wykorzystując w tym celu opisane „modele elementarne”. Model szumowy przyrządu jest ściśle związany z jego małosygnalowym modelem liniowym. Omówienie szczegółów budowy modelu szumowego każdego przyrządu z osobna odłożymy jednak do rozdziału 6.

### 3.2.2. Instrukcja analizy szumów

Składnia instrukcji zlecającej programowi PSpice wykonanie analizy szumów jest następująca:

**.NOISE \_wyjście \_źródło \_krok**

Przykład:

.NOISE V(5) VIN 10

Instrukcja analizy szumów musi być poprzedzona instrukcją analizy zmiennoprądowej .AC — patrz strona 45. Analiza szumów dokonywana jest w tym samym przedziale częstotliwości, w którym dokonywana jest małosygnalowa analiza zmiennoprądowa. Parametr **\_krok** oznacza co ile punktów na osi częstotliwości, w stosunku do analizy AC, będzie wykonywana analiza szumów. Jeżeli parametr ten zostanie pominięty lub zostanie mu nadana wartość zero to obliczenia nie będą wykonywane.

Program PSpice oblicza:

- Gęstość widmowa szumów, wyrażoną w  $[V]/[Hz]^{\frac{1}{2}}$ , na wyjściu napięciowym układu, które określone jest przez parametr **\_wyjście**.
- Gęstość widmowa szumów, wyrażoną w  $[V]/[Hz]^{\frac{1}{2}}$  lub w  $[A]/[Hz]^{\frac{1}{2}}$ , zredukowaną do wejścia układu; przez wejście układu rozumie się niezależne źródło napięcia lub prądu, którego nazwa określona jest przez parametr **\_źródło**.

Sposób wyprowadzania wyników analizy szumowej za pomocą instrukcji .PRINT i .PLOT został przedstawiony, przy okazji prezentacji tych instrukcji (patrz strony 52, 55). W przypadku stosowania instrukcji .PROBE, wielkości, które można

**Tablica III** Parametry tranzystorów układu  $\mu A741$ .

	NPN	PNP	Jednostki
BF	80	10	-
BR	1	1	-
IS	$10^{-14}$	$10^{-14}$	[A]
RB	100	20	[ $\Omega$ ]
VA	50	50	[V]
TF	0.3	0.3	[ns]
TR	6.0	20.0	[ns]
CCS	2.0	0	[pF]
CJE	3.0	6.0	[pF]
CJC	2.0	4.0	[pF]
KF	$6.6 \cdot 10^{-16}$	$3 \cdot 10^{-12}$	$[A]^{(1-AF)}/[s]$
AF	1	1.5	-

wyprowadzić to:

- V(ONoise)** gęstość widmowa napięcia szumów na wyjściu układu;
- V(INoise)** gęstość widmowa napięcia szumów zredukowana do wejścia układu w przypadku, gdy na wejściu układu znajduje się źródło napięcia;
- I(INoise)** gęstość widmowa prądu szumów zredukowana do wejścia układu w przypadku, gdy na wejściu znajduje się źródło prądu.

Program PROBE, do którego przekazywane są te dane, pozwala dalej na obliczenie np. całki z funkcji gęstości napięcia szumów. Daje to w rezultacie funkcję obrazującą całkowite napięcie szumów na układzie jako funkcję częstotliwości górnej przedziału częstotliwości.

### 3.2.3. Szumy wtórnika napięcia — instrukcje .INC, .SUBCKT, .ENDS

Dany jest układ wtórnika napięciowego przedstawiony na Rys. 33. Wtórnik zbudowano w oparciu o wzmacniacz operacyjny  $\mu A741$ . Przyjmijmy, że schemat elektryczny układu  $\mu A741$  jest taki sam jak w podręczniku W.Marciniaka [20] — Rys. 34. Tablica III zawiera wartości parametrów tranzystorów bipolarnych [24], które należy przyjąć do obliczeń.

- Obliczyć gęstość widmową napięcia szumów na wyjściu układu.
- Obliczyć całkowite napięcie szumów na wyjściu układu jako funkcję maksymalnej częstotliwości. Przyjąć, że przedział częstotliwości rozciąga się od 20Hz do 2MHz.
- Obliczyć gęstość widmową szumów napięciowych i prądowych zredukowanych do wejścia układu.

Dane dla programu PSpice niezbędne do przeprowadzenia obliczeń są przedstawione poniżej.

```
ANALIZA SZUMÓW UKŁADU UA741
*szumy napięciowe
.INC UA741.CIR
```

```
VCC+ 4 0 15V
VCC- 5 0 -15V
```

```
R2 2 3 1K
```


```
XOPAMP 1 2 4 5 3 UA741N
RIN 7 1 1K
VIN 7 0 AC 1 0
```

```
.AC DEC 20 20 2MEG
.NOISE V(3) VIN
.PROBE
.END
```

```
ANALIZA SZUMÓW UKŁADU UA741
*szumy prądowe
.INC UA741.CIR
```

```
VCC+ 4 0 15V
VCC- 5 0 -15V
```

```
R2 2 3 1K
XOPAMP 1 2 4 5 3 UA741N
RIN 7 1 1K
IIN 7 0 AC 1
```



Rys.33. Wtórnik napięciowy.

```
.AC DEC 20 20 2MEG
.NOISE V(3) IIN
.PROBE
.END
```

Schemat elektryczny wzmacniacza operacyjnego  $\mu A741$  zadeklarowany został w odrębnym zbiorze i dołączony za pomocą instrukcji `.INC` (ang. include — dołącz). Postać ogólna tej instrukcji jest następująca:


**.INC nazwa\_zbioru**

Przykład:

```
.INC ..\LIB\OBWOD.MOJ
```

Po natrafieniu na tę instrukcję program PSpice dołącza do zbioru danych wejściowych, w miejsce instrukcji `.INC`, zawartość zbioru tekstowego, którego nazwa określona jest przez parametr **nazwa\_zbioru**. W zbiorze tym mogą się znaleźć dowolne deklaracje i instrukcje akceptowane przez program PSpice. W szczególności, może się tam znaleźć kolejna instrukcja `.INC`. Nazwę dołączanego zbioru można poprzedzić pełną ścieżką dostępu. Instrukcja `.INC` dostępna jest tylko w najnowszych wersjach programu PSpice.

W naszym przypadku dołączany zbiór o nazwie UA741.CIR znajdował się w tej samej kartotece, w której znajdował się zbiór zawierający dane wejściowe. Zawartość dołączanego zbioru pokazana jest poniżej.



Rys.34. Schemat elektryczny wzmacniacza operacyjnego  $\mu A741$ .

```

Q18 18 19 20 TPNP
Q19 14 20 23 TPNP
Q20 9 22 14 TPNP
.MODEL TPNP NPN BF=80 BR=1 IS=1.0E-14 RB=100 VA=50 TF=0.3N TR=6.0N
+      CCS=2.0P CJE=3.0P CJC=2.0P KF=6.6E-16 AF=1
.MODEL TPNP PNP BF=10 BR=1 IS=1E-14 RB=20 VA=50 TF=0.3N TR=20.0N
+      CCS=0.0P CJE=6.0P CJC=4.0P KF=3.06E-12 AF=1.5
.ENDS

```

Deklaracja struktury wzmacniacza operacyjnego zamknięta została w postaci podobwodu. Wykorzystano w tym celu instrukcję `.SUBCKT` (ang. **subcircuit** — podobwód). Składnia tej instrukcji jest następująca:

**`.SUBCKT _nazwa _n1 [_n2 _n3 ...]`**

Przykład:

```
.SUBCKT LM311 1 2 3 4
```

Instrukcja `.SUBCKT` rozpoczyna deklarację struktury podobwodu (wielobiegunnika). Nazwa podobwodu określona jest przez parametr `_nazwa`. Numery węzłów wewnętrznych udostępnianych na zewnątrz podobwodu to `_n1`, `_n2`, `_n3`, ... . Linie następujące po instrukcji `.SUBCKT` zawierać powinny instrukcje i deklaracje definiujące strukturę podobwodu. Zabronione jest używanie w tym miejscu instrukcji sterujących analizą obwodu. Dozwolone jest natomiast zdefiniowanie kolejnego podobwodu, wywołanie istniejącego już podobwodu, zadeklarowanie i wywołanie modelu przyrządu półprzewodnikowego lub elementu pasywnego. Należy przy tym podkreślić, że wszelkie nazwy definiowane wewnątrz podobwodu mają znaczenie *lokalne* tzn. nie są rozpoznawane przez program PSpice na zewnątrz podobwodu. Dotyczy to także węzłów, których numery mają w przypadku podobwodu znaczenie *lokalne* z wyjątkiem węzła masy, którego numer (0) ma zawsze znaczenie *globalne*. Deklaracja struktury podobwodu kończy się instrukcją `.ENDS` (ang. **end** of **subcircuit** — koniec podobwodu). Składnia tej instrukcji jest następująca:

**`.ENDS [_nazwa]`**

Przykład:

```
.ENDS LM311
```

Instrukcja `.ENDS` informuje program PSpice, że została zakończona definicja struktury podobwodu o nazwie określonej przez parametr `_nazwa`. Jeżeli parametr ten zostanie pominięty to zakończone zostaną wszystkie rozpoczęte definicje struktury podobwodu. W praktyce parametr `_nazwa` podawany jest tylko w przypadku zagnieżdżania się definicji podobwodów.


Zdefiniowaną strukturę podobwodu wywołać można posługując się quasi–elementem, którego nazwa zaczyna się od litery X.

**`Xyyyyyyy _n1 [_n2 _n3 ...] _nazwa`**

Przykład:

```
XOPA1 4 12 6 7 9 LM311
```






**Rys.35.** Gęstość widmowa napięcia szumów na wyjściu układu wtórnika (A) i całkowite napięcie szumów wyjściowych (B).

Parametry **\_n1**, **\_n2**, **\_n3** określają numery węzłów obwodu, w które wpięty zostanie podobwód. Nazwa wpinanej struktury określona jest przez parametr **\_nazwa**. Należy zwrócić przy tym uwagę, że *dozwolone są odwołania w przód*. Oznacza to, że w strukturę obwodu można wpiąć podobwód, którego struktura zdefiniowana zostanie dopiero w późniejszej kolejności.

W tej chwili znamy już składnię wszystkich instrukcji i deklaracji użytych do zdefiniowania struktury wtórnika z Rys. 33. Należy zwrócić uwagę, że w zbiorze danych wejściowych znalazł się opis dwóch obwodów. Pierwszy z nich pozwala na obliczenie gęstości widmowej napięcia szumów na wyjściu układu oraz gęstość widmową napięcia szumów zredukowaną do wejścia (źródło VIN). W drugim obwodzie niezależne źródło napięcia wpięte na wejściu oryginalnego obwodu zastąpione zostało przez źródło prądu IIN. W ten sposób w wyniku analizy szumów otrzymać można gęstość widmową prądu szumów zredukowaną do wejścia układu (IIN). Umieszczenie deklaracji obu obwodów w jednym zbiorze wejściowym pozwoliło na wyświetlenie, za pomocą programu PROBE, gęstości widmowej napięcia i prądu szumów zredukowanych do wejścia na jednym wspólnym wykresie Rys. 36.

Obliczenie całkowitego napięcia szumów na wyjściu układu wymaga scałkowania gęstości widmowej napięcia szumów na wyjściu Rys. 35a. Można się posłużyć w tym celu funkcją  $s()$



**Rys.36.** Gęstość widmowa napięcia i prądu szumów na wejściu układu wtórnika.

dostępna w programie PROBE. Jest ona określona wzorem:

$$s[w(x)] = \int_0^x w(x) dx \quad (48)$$

gdzie:

$x$  zmienna opisująca oś na wykresie poziomą;

$w(x)$  dowolne akceptowane przez program Probe wyrażenie (patrz strona 217).

Wykres obrazujący całkowite napięcie wyjściowe szumów jako funkcję maksymalnej rozważanej częstotliwości (minimalna częstotliwość wynosi 20Hz) przedstawia Rys. 35b. Na tym samym wykresie przedstawiono funkcję gęstości napięcia szumów wyjściowych. Dla niskich częstotliwości widoczny jest wyraźnie składnik  $1/f$  (szumy migotania). Dla częstotliwości wysokich zmniejszenie gęstości widmowej szumów związane jest ze zmniejszeniem wzmocnienia układu poniżej wartości 1.0 — należy się spodziewać, że obserwowane szumy powstają głównie w pobliżu wejścia wzmacniacza operacyjnego i wzmocniane są podobnie jak sygnał wejściowy układu.